# $Id: PKGBUILD 64990 2012-02-19 12:05:01Z seblu $
# Maintainer: Sébastien Luttringer <seblu@aur.archlinux.org>
# Contributor: eric <eric@archlinux.org>
# Contributor: Tom Newsom <Jeepster@gmx.co.uk>
# Contributor: Chris Brannon <cmbrannon79@gmail.com>

pkgname=stow
pkgver=2.2.0
pkgrel=1
pkgdesc='Manage installation of localy builded softwares'
url='http://www.gnu.org/software/stow/'
arch=('any')
source=("ftp://ftp.gnu.org/pub/gnu/$pkgname/$pkgname-$pkgver.tar.gz")
license=('GPL2')
makedepends=('texi2html')
checkdepends=('perl-test-output')
install=$pkgname.install
md5sums=('902564d502975f353d3b0514a21863b1')

build() {
  cd $pkgname-$pkgver
  ./configure --prefix=/usr --mandir=/usr/share/man --infodir=/usr/share/info \
    --with-pmdir=/usr/lib/perl5/vendor_perl/
  make
}

check() {
  cd $pkgname-$pkgver
  make check
}

package() {
  cd $pkgname-$pkgver
  make DESTDIR="$pkgdir" install
}

# vim:set ts=2 sw=2 ft=sh et:
